




# DECUS

## PROGRAM LIBRARY

DECUS NO.	8-304
TITLE	PSEUDO-NOISE (P-N) SEQUENCE TEST
AUTHOR	R. G. Smith
COMPANY	Carleton University Faculty of Engineering Ottawa, Ontario, Canada
DATE	August 1969
SOURCE LANGUAGE	PAL



# PSEUDO-NOISE (P-N) SEQUENCE TEST

DECUS Program Library Write-up

DECUS No. 8-304

## ABSTRACT

This program can be used to determine the statistical characteristics of a pseudo-random sequence generator. The probability density function can be studied with the aid of an amplitude histogram, and measures of the "randomness" can be determined with correlation diagrams, sample function displays, and scattergrams. All diagrams are displayed on the CRT.

## REQUIREMENTS

### Storage

Program: 0-1,16-171,200-356,400-572,600-713

Date: 2000-5777\*

Axis Points: 1000-1777\*

\*Data and Axis Points are stored as (x, y) co-ordinates, (x<sub>0</sub>y<sub>0</sub>x<sub>1</sub>y<sub>1</sub>.....)

### Equipment

PDP-8 with EAE, display, A/D converter, and external interrupt. The 6424 instruction seen in this program resets the clock interrupt at the Carleton installation. It is used to time sampling of the pseudo-random sequences and should be replaced by the appropriate instruction or NOP when implemented at a different installation.

## USAGE

### Calling Sequence

The program is called by setting the switch register to 0200 and pressing "load address." The switches are reset for the desired test configuration and "start" is pressed.

### Switch Settings

(I) The type of test to be used is determined by switches 1 and 2.

S <sub>1</sub>	S <sub>2</sub>	
0	0	Amplitude Histogram
0	1	Correlation Diagram
1	0	Scattergram
1	1	Sample Function Display



(II) The input mode is determined by switch 0.

$S_0$	
0	Software*
1	A/D converter (channel 0)

\*Unless changes are made, the program will extract pseudo-random numbers from its own reference software generator, RAND, stored in locations 0136-0156. It is a mixed congruential generator whose sequence is defined by the following algorithm.

$$x_{i+1} = x_i(2^6+3) + 3 \quad \text{modulo } 2^{12}$$

where  $x_0=1$ .

If a different software generator is required, it must be read into unused locations in memory and a JMS to its starting address inserted in location 0074.

(III) When the correlation diagram is used, the shift (minus one) between the samples plotted as co-ordinate pairs is determined by switches 3-11. Hence, the maximum shift is 512.

(IV) A change in the test being run can be effected by resetting the switch register for the new test, and depressing any teletype key.

## RESTRICTIONS

### Allowable Range

This program is intended for testing 10 bit pseudo-random numbers, considered to be in the low order 10 bits of the accumulator.

## DESCRIPTION

### Discussion

(I) Amplitude Histogram - This is a free running accumulation of the number of occurrences of a particular number in a finite record of a pseudo-random sequence. The maximum accumulation is determined by the rise increment, IN. Each time a particular number occurs, its accumulation counter is incremented by IN. With IN set at  $10_8=8_{10}$ , the maximum accumulation is  $177_8 = 128_{10}$ . A visual indication of the first order probability density function of the sequence under is given by this display.

For pseudo-random number generators with uniform distributions, the mean accumulation and variance (on the mean accumulation) may be determined as follows:

Stop the histogram test when a suitable display has been obtained (an approximate mean accumulation more than half way up the vertical axis). The mean accumulation will appear in the ACC, as defined by:

$$\bar{y}_n = 1/1024 \sum_{i=1}^{1024} y_i$$

where  $y_i$  is the value of the  $i^{\text{th}}$  accumulation counter. This must be normalized by  $N$ , the number of samples, to be compared with the theoretical estimate (assuming  $N$  independent trials of the generator).

$$\bar{y}_t = 1/1024 \quad (\text{for } 1024 \text{ quantization levels})$$

Press "continue." The variance on the mean accumulation will appear, high order 12 bits in the ACC, low order 12 bits in the MQ:

$$\sigma^2 = \sum_{i=1}^{1024} (y_i - \bar{y}_n)^2$$

This must be divided by  $(1024 \cdot N^2)$  for proper normalization. It may then be compared with the theoretical estimate of the variance on the mean accumulation (again assuming  $N$  independent trials).

$$\sigma_t^2 = 1/(1024 \cdot N)$$

Press "continue." The number of samples taken for the histogram will appear, high order 12 bits in the ACC, low order 12 bits in the MQ.

(II) Correlation Diagram - A visual indication of the correlation between successive samples may be obtained from the correlation diagram. It is a display of  $(1024+(s+1))$  samples. Co-ordinates are formed from  $(r_i, r_{i+s+1})$  where the  $r_i$  are samples of the pseudo-random sequence.

(III) Scattergram - A measure of the "randomness" of the sequence under test may be obtained from the scattergram or successive pairs diagram. Co-ordinates are formed by plotting successive pairs of samples of the sequence as  $(x, y)$  co-ordinates as follows:

$$(r_0, r_1), (r_2, r_3), (r_4, r_5) \dots\dots\dots$$

This display is free-running and takes in 1024 such pairs before overwriting the oldest pairs. Further details of the use of this diagram are given by Chambers (1).

(IV) Sample Function Display - Further indications of the structure or patterns in the sequence are provided by the display of a finite record of the output of the generator under test. In this display,  $(i, r_i)$  is plotted as an  $(x, y)$  co-ordinate, where  $r_i$  is a sample of the pseudo-random sequence, and  $i$  is the sample index. The display is free-running and takes in 1024 samples before overwriting the oldest samples.



## EXECUTION TIME

The histogram test has an approximate maximum frequency of 6 khz. The other tests have approximate maximum interrupt frequencies of 4 khz.

These frequencies are quoted for sampling from an A/D converter with a 50 usec. conversion time.

## REFERENCES

1. R. P. Chambers, "Random-number Generation On Digital Computers," IEEE Spectrum, February 1967, pp. 48-56.

/PSEUDO-RANDOM SEQUENCE GENERATOR  
 /AMPLITUDE HISTOGRAM, CORRELATION  
 /DIAGRAM (SCATTERGRAM), AND SAMPLE  
 /FUNCTION OBSERVATION.

		*0000
0000	0000	0000
0001	5100	JMP SER / JUMP TO INTERRUPT SERVICE
		*0016
0016	0000	MEM,0
0017	0000	MEMI,0
0020	0000	N0,0
0021	0000	N01,0
0022	0000	XMIN,0
0023	0000	MTEMP,0
0024	0000	X,0
0025	0000	Y,0
0026	0400	THOU,0400 / NUMBER OF AXIS POINTS
0027	7600	CNT, 7600 / NO. OF POINTS ON 1 AXIS
0030	0000	CNT1,0
0031	0010	INC,0010 / INCREMENT BETWEEN POINTS
0032	0000	A0,0 / ORIGIN
0033	0000	B0,0 / SETTINGS
0034	0000	AC,0 / ACC-TEMPORARY STORAGE
0035	0000	PL1,0000
0036	0200	PL2,0200
0037	0777	M0,0777 / INITIAL AXIS ADDRESS MINUS ONE
0040	2000	X0,2000 / INITIAL X ADDRESS
0041	2001	Y0,2001 / INITIAL Y ADDRESS
0042	0000	XLOC,0
0043	0000	YLOC,0
0044	0000	CNTR,0
0045	2000	NRAND,2000
0046	0000	IR,0 / SAMPLE INDEX
0047	0000	S,0 / SEPARATION MINUS ONE
0050	0000	T1,0
0051	1777	MASK,1777
0052	0777	MASKS,0777
0053	0177	MASKV,0177
0054	0003	MASKH,0003
0055	0535	EXEC,AXIS
0056	1000	CORR,1000 / Y AXIS SHIFT ON HISTOGRAM
0057	0336	HT,HSTINT
0060	0400	SC1,SCAT1 / INTERRUPT SERVICE POINTERS
0061	0417	SC2,SCAT2
0062	0442	SC3,SCAT3
0063	0456	SC4,WAIT
0064	0460	SCP,SEQ
0065	0507	SA1,SAMP1
0066	7777	SOFT,7777
0067	0000	HIGH,0000



0070	0000	LOW, 0000
0071	0000	HOR, 0000
0072	0000	LOR, 0000
0073	4157	CHGE, JMS A2D / SOFTWARE, A/D CONVERTER
0074	4136	SET, JMS RAND / SAMPLING LINKS
0075	0000	CNIX, 0
0076	0000	NPT1, 0 / SAMPLES COUNTER, HIGH ORDER
0077	0000	NPT2, 0 / LOW ORDER
0100	3034	SER, DCA AC / INTERRUPT SERVICE LINK
0101	6424	6424 / RESET CLOCK
0102	6031	6031 / KEYBOARD INTERRUPT?
0103	74T0	SKP
0104	5436	JMP I PL2 / YES, SET UP NEW TEST
0105	4435	JMS I PL1 / NO, SERVICE INTERRUPT
0106	7300	CLA CLL
0107	1034	TAD AC / RESTORE ACC
0110	6001	ION / ENABLE INTERRUPT
0111	5400	JMP I 0000 / CONTINUE DISPLAY
0112	6077	TRACE, 6077 / DISPLAY ROUTINE-SET INTENSITY
0113	6424	6424 / RESET CLOCK
0114	6001	ION / ENABLE INTERRUPT
0115	7300	START, CLA CLL
0116	1016	TAD MEM / PUT INITIAL ADDRESS MINUS
0117	3023	DCA MTEMP / ONE IN TEMP STORE
0120	1020	TAD N0
0121	7041	CIA / SET POINTS COUNTER
0122	3021	DCA N01
0123	1416	C, TAD I MEM
0124	6052	6052
0125	7300	CLA CLL
0126	1416	TAD I MEM / DISPLAY CO-ORDINATE
0127	6066	6066
0130	7300	CLA CLL
0131	2021	ISZ N01 / FINISHED?
0132	5123	JMP C / NO, CONTINUE
0133	1023	TAD MTEMP / YES, REINITIALIZE
0134	3016	DCA MEM
0135	5115	JMP START / RESTART DISPLAY
0136	0000	RAND, 0 / MIXED CONGRUENTIAL
0137	7621	CLA MQL / PSEUDO-RANDOM NUMBER
0140	1155	TAD RN / GENERATOR
0141	7413	SHL / $R(I+1)=R(I)*(2 \uparrow 6+3)+3$
0142	0005	0005 / MODULO $2 \uparrow 12$
0143	1155	TAD RN
0144	1155	TAD RN
0145	1155	TAD RN
0146	1156	TAD THREE
0147	3155	DCA RN
0150	1155	TAD RN
0151	7012	RTR
0152	0051	AND MASK / TAKE HIGH ORDER 10 BITS
0153	5536	JMP I RAND / FOR DISPLAY TESTS



Ø154	ØØØ1	RNØ, ØØØ1
Ø155	ØØØØ	RN, ØØØØ
Ø156	ØØØ3	THREE, ØØØ3
Ø157	ØØØØ	A2D, Ø / SAMPLE SEQUENCE VIA A/D CONVERTER
Ø16Ø	73ØØ	CLA CLL
Ø161	6542	ADSC / SET TO CHANNEL Ø
Ø162	6532	ADCV
Ø163	6531	ADSF
Ø164	5163	JMP .-1 / SAMPLE SEQUENCE
Ø165	6534	ADRB
Ø166	5557	JMP I A2D / RETURN
Ø167	ØØØØ	FETCH, Ø / FETCH A SAMPLE
Ø17Ø	4136	G1, JMS RAND
Ø171	5567	JMP I FETCH / RETURN
		*Ø2ØØ / START OF TESTS
Ø2ØØ	6Ø32	6Ø32 / CLEAR KEYBOARD FLAG
Ø2Ø1	73ØØ	CLA CLL
Ø2Ø2	1154	TAD RNØ / INITIALIZE SOFTWARE
Ø2Ø3	3155	DCA RN / GENERATOR
Ø2Ø4	1Ø74	TAD SET
Ø2Ø5	317Ø	DCA G1 / SET SOFTWARE-A/D CONVERTER CHOICE
Ø2Ø6	1Ø37	TAD MQ / SET INITIAL DATA STORAGE ADDRESS
Ø2Ø7	3Ø16	DCA MEM
Ø21Ø	1Ø45	TAD NRAND / SET POINTS COUNTER
Ø211	1Ø26	TAD THOU
Ø212	3Ø2Ø	DCA NØ
Ø213	76Ø4	LAS
Ø214	71Ø4	CLL RAL
Ø215	762Ø	SNL CLA / READ BIT Ø; CHOOSE INPUT MODE
Ø216	5221	JMP TSTRD / SOFTWARE REQUIRED (Ø)
Ø217	1Ø73	TAD CHGE / A/D CONVERTER REQUIRED (1)
Ø22Ø	317Ø	DCA G1
Ø221	76Ø4	TSTRD, LAS
Ø222	7ØØ6	RTL
Ø223	7ØØ6	RTL
Ø224	ØØ54	AND MASKH / READ BITS 1,2; CHOOSE TEST
Ø225	745Ø	SNA
Ø226	5236	JMP HIS / HISTOGRAM REQUIRED (ØØ)
Ø227	1Ø66	TAD SOFT
Ø23Ø	745Ø	SNA
Ø231	5244	JMP SCA1 / SCATTERGRAM REQUIRED (Ø1)
Ø232	1Ø66	TAD SOFT
Ø233	765Ø	SNA CLA
Ø234	5257	JMP SCAP / SUCCESSIVE PAIRS PLOT (Ø2)
Ø235	5265	JMP SCA2 / SAMPLE FUNCTION REQUIRED (Ø3)
Ø236	1Ø56	HIS, TAD CORR / HISTOGRAM PRELIMINARY
Ø237	3Ø22	DCA XMIN / CORRECT Y AXIS MARKER
Ø24Ø	3Ø24	DCA X / CLEAR AXIS COUNTER
Ø241	1Ø57	TAD HT / SET UP HISTOGRAM INTERRUPT

Ø242	3Ø35		DCA PL1 / SERVICE LINK
Ø243	5316		JMP HISTGM / PROCEED WITH HISTOGRAM
Ø244	4272	SCA1,	JMS CLEAR / SCATTERGRAM PRELIMINARY
Ø245	1Ø6Ø		TAD SCI / SET UP SCATTERGRAM INTERRUPT
Ø246	3Ø35		DCA PL1 / SERVICE LINK
Ø247	76Ø4		LAS / DETERMINATION SEPARATION (S)
Ø25Ø	ØØ52		AND MASKS / FROM BITS 3-11
Ø251	7ØØ1		IAC
Ø252	3Ø47		DCA S
Ø253	1Ø47		TAD S
Ø254	7Ø41		CIA / SET HISTORY COUNTER
Ø255	3Ø44		DCA CNTR
Ø256	5455		JMP I EXEC / JUMP TO AXIS GENERATION
Ø257	4272	SCAP,	JMS CLEAR / SUCCESSIVE PAIRS PLOT
Ø26Ø	1Ø64		TAD SCP / PRELIMINARY
Ø261	3Ø35		DCA PL1
Ø262	724Ø		CLA CMA / SET XY COUNTER
Ø263	3Ø75		DCA CNTX
Ø264	5455		JMP I EXEC / JUMP TO AXIS GENERATION
Ø265	4272	SCA2,	JMS CLEAR / SAMPLE FUNCTION PRELIMINARY
Ø266	1Ø65		TAD SAI / SET UP SAMPLE FUNCTION INTERRUPT
Ø267	3Ø35		DCA PL1 / SERVICE LINK
Ø27Ø	3Ø46		DCA IR / CLEAR SAMPLE INDEX
Ø271	5455		JMP I EXEC / JUMP TO AXIS GENERATION
Ø272	ØØØØ	CLEAR,	Ø / INITIALIZATION
Ø273	1Ø45		TAD NRAND
Ø274	71Ø4		CLL RAL
Ø275	7Ø41		CIA
Ø276	3Ø21		DCA NØ1
Ø277	1Ø4Ø		TAD XØ
Ø3ØØ	3Ø42		DCA XLOC
Ø3Ø1	3442		DCA I XLOC / CLEAR DATA REGISTERS
Ø3Ø2	2Ø42		ISZ XLOC
Ø3Ø3	2Ø21		ISZ NØ1
Ø3Ø4	53Ø1		JMP .-3
Ø3Ø5	1Ø4Ø		TAD XØ / CLEAR X ADDRESS
Ø3Ø6	3Ø42		DCA XLOC
Ø3Ø7	1Ø41		TAD YØ
Ø31Ø	3Ø43		DCA YLOC / CLEAR Y ADDRESS
Ø311	3Ø22		DCA XMIN / CLEAR Y AXIS MARKER
Ø312	1Ø45		TAD NRAND
Ø313	7Ø41		CIA / SET POINTS COUNTER
Ø314	3Ø5Ø		DCA TI
Ø315	5672		JMP I CLEAR / RETURN
Ø316	1Ø45	HISTGM,	TAD NRAND / HISTOGRAM SETUP
Ø317	7Ø41		CIA / SET POINTS COUNTER (1Ø24)
Ø32Ø	3Ø21		DCA NØ1
Ø321	1Ø4Ø		TAD XØ
Ø322	3Ø42		DCA XLOC / SET REGISTERS



Ø323	1Ø24	CLR,	TAD X
Ø324	3442		DCA I XLOC
Ø325	2Ø42		ISZ XLOC
Ø326	3442		DCA I XLOC
Ø327	2Ø42		ISZ XLOC
Ø33Ø	2Ø24		ISZ X
Ø331	2Ø21		ISZ NØ1
Ø332	5323		JMP CLR
Ø333	3Ø76		DCA NPT1 / CLEAR SAMPLES COUNTER
Ø334	3Ø77		DCA NPT2
Ø335	5455		JMP I EXEC / JUMP TO AXIS GENERATION
Ø336	ØØØØ	HSTINT,	Ø / HISTOGRAM INTERRUPT SERVICE
Ø337	73ØØ		CLA CLL
Ø34Ø	1Ø77		TAD NPT2
Ø341	7ØØ1		IAC / INCREMENT SAMPLES COUNTER (24 BIT)
Ø342	3Ø77		DCA NPT2
Ø343	7ØØ4		RAL
Ø344	1Ø76		TAD NPT1
Ø345	3Ø76		DCA NPT1
Ø346	4167		JMS FETCH / FETCH SAMPLE
Ø347	71Ø4		CLL RAL / TIMES TWO FOR PROPER STORAGE
Ø35Ø	1Ø41		TAD YØ
Ø351	3Ø43		DCA YLOC
Ø352	1443		TAD I YLOC / UPDATE DISPLAY
Ø353	1356		TAD IN
Ø354	3443		DCA I YLOC
Ø355	5736		JMP I HSTINT / RETURN
Ø356	ØØ1Ø	IN,ØØ1Ø	/ RISE INCREMENT OF HISTOGRAM *Ø4ØØ
Ø4ØØ	ØØØØ	SCAT1,	Ø / SCATTERGRAM SERVICE 1
Ø4Ø1	2Ø44		ISZ CNTR
Ø4Ø2	741Ø		SKP
Ø4Ø3	5211		JMP FIN / TAKE IN S SAMPLES AND STORE
Ø4Ø4	4167	F1,	JMS FETCH / IN X ARRAY
Ø4Ø5	3442		DCA I XLOC
Ø4Ø6	2Ø42		ISZ XLOC
Ø4Ø7	2Ø42		ISZ XLOC
Ø41Ø	56ØØ		JMP I SCAT1
Ø411	1Ø61	FIN,	TAD SC2 / FINISHED. CHANGE INTERRUPT
Ø412	3Ø35		DCA PL1 / LINK
Ø413	1Ø5Ø		TAD T1
Ø414	1Ø47		TAD S
Ø415	3Ø5Ø		DCA T1
Ø416	52Ø4		JMP F1
Ø417	ØØØØ	SCAT2,	Ø / SCATTERGRAM SERVICE 2
Ø42Ø	2Ø5Ø		ISZ T1
Ø421	741Ø		SKP
Ø422	5234		JMP FIN2 / TAKE IN (NRAND-S)
Ø423	4167	F2,	JMP FETCH / SAMPLES AND STORE
Ø424	3443		DCA I YLOC / IN X AND Y ARRAYS

Ø425	1443		TAD I YLOC
Ø426	3442		DCA I XLOC
Ø427	2Ø42		ISZ XLOC
Ø43Ø	2Ø42		ISZ XLOC
Ø431	2Ø43		ISZ YLOC
Ø432	2Ø43		ISZ YLOC
Ø433	5617		JMP I SCAT2
Ø434	1Ø62	FIN2,	TAD SC3 / FINISHED. CHANGE INTERRUPT
Ø435	3Ø35		DCA PL1 / LINK
Ø436	1Ø47		TAD S
Ø437	7Ø41		CIA
Ø44Ø	3Ø44		DCA CNTR
Ø441	5223		JMP F2
Ø442	ØØØØ	SCAT3,	Ø / SCATTERGRAM SERVICE 3
Ø443	2Ø44		ISZ CNTR
Ø444	741Ø		SKP
Ø445	5253		JMP FIN3 / TAKE IN S SAMPLES AND
Ø446	4167	F3,	JMS FETCH / STORE IN Y ARRAY
Ø447	3443		DCA I YLOC
Ø45Ø	2Ø43		ISZ YLOC
Ø451	2Ø43		ISZ YLOC
Ø452	5642		JMP I SCAT3
Ø453	1Ø63	FIN3,	TAD SC4 / FINISHED. CHANGE INTERRUPT
Ø454	3Ø35		DCA PL1 / LINK
Ø455	5246		JMP F3
Ø456	ØØØØ	WAIT,	Ø / INTERRUPT WAIT LOOP
Ø457	5656		JMP I WAIT
Ø46Ø	ØØØØ	SEQ,	Ø / SUCCESSIVE PAIRS SERVICE
Ø461	4167		JMS FETCH / FETCH SAMPLE
Ø462	2Ø75		ISZ CNTX / X OR Y
Ø463	5272		JMP YST
Ø464	3442		DCA I XLOC / STORE X
Ø465	2Ø42		ISZ XLOC
Ø466	2Ø42		ISZ XLOC
Ø467	7344		CLA CLL CMA RAL / RESET COUNTER
Ø47Ø	3Ø75		DCA CNTX
Ø471	566Ø		JMP I SEQ / RETURN
Ø472	3443	YST,	DCA I YLOC / STORE Y
Ø473	2Ø43		ISZ YLOC
Ø474	2Ø43		ISZ YLOC
Ø475	2Ø5Ø		ISZ T1 / ADDRESS RESET NEEDED?
Ø476	566Ø		JMP I SEQ / NO, RETURN
Ø477	1Ø4Ø		TAD XØ / YES
Ø5ØØ	3Ø42		DCA XLOC
Ø5Ø1	1Ø41		TAD YØ
Ø5Ø2	3Ø43		DCA YLOC
Ø5Ø3	1Ø45		TAD NRAND
Ø5Ø4	7Ø41		CIA
Ø5Ø5	3Ø5Ø		DCA T1



0506	5660		JMP I SEQ / RETURN
0507	0000	SAMP1,	0 / SAMPLE FUNCTION INTERRUPT SERVICE
0510	7300		CLA CLL
0511	2050		ISZ TI
0512	5323		JMP GG3
0513	1045		TAD NRAND / RESET ADDRESSES
0514	7041		CIA
0515	3050		DCA TI
0516	1040		TAD X0
0517	3042		DCA XLOC
0520	3046		DCA IR
0521	1041		TAD Y0
0522	3043		DCA YLOC
0523	4167	GG3,	JMS FETCH / FETCH SAMPLE
0524	3443		DCA I YLOC
0525	2043		ISZ YLOC / UPDATE DISPLAY
0526	2043		ISZ YLOC
0527	1046		TAD IR
0530	3442		DCA I XLOC
0531	2042		ISZ XLOC
0532	2042		ISZ XLOC
0533	2046		ISZ IR
0534	5707		JMP I SAMP1 / RETURN
0535	1016	AXIS,	TAD MEM / AXIS GENERATION
0536	3023		DCA MTEMP
0537	4365		JMS RSET / INITIALIZE
0540	1022	A,	TAD XMIN
0541	3416		DCA I MEM
0542	1025		TAD Y / STORE VERTICAL AXIS POINTS
0543	3416		DCA I MEM
0544	1025		TAD Y
0545	1031		TAD INC / INCREMENT COUNTER
0546	3025		DCA Y
0547	2030		ISZ CNT1
0550	5340		JMP A
0551	4365		JMS RSET / INITIALIZE
0552	1024	B,	TAD X
0553	3416		DCA I MEM / STORE HORIZONTAL AXIS POINTS
0554	1024		TAD X
0555	1031		TAD INC / INCREMENT COUNTER
0556	3024		DCA X
0557	3416		DCA I MEM
0560	2030		ISZ CNT1
0561	5352		JMP B
0562	1023		TAD MTEMP
0563	3016		DCA MEM
0564	5112		JMP TRACE / JUMP TO DISPLAY
0565	0000	RSET,	0 / INITIALIZATION ROUTINE
0566	1027		TAD CNT

0567	3030		DCA CNTI
0570	3024		DCA X
0571	3025		DCA Y
0572	5765		JMP I RSET
			*0600
0600	4247	MEAN,	JMS INIT / CALCULATE SAMPLE MEAN AND VARIANCE
0601	1443	GO,	TAD I YLOC / ON HISTOGRAM
0602	7010		RAR
0603	7012		RTR / CORRECT FOR HISTOGRAM RISE INCREMENT
0604	0053		AND MASKV / 0177 IS MAX ACCUMULATION (128)
0605	3070		DCA LOW
0606	4263		JMS SUM / SUM ACCUMULATIONS
0607	2043		ISZ YLOC / AND COMPUTE MEAN
0610	2043		ISZ YLOC
0611	2021		ISZ N01
0612	5201		JMP GO
0613	4304		JMS AV / AVERAGE AND DISPLAY MEAN IN ACC
0614	7402		HLT
0615	7041	VAR,	CIA / INVERT & STORE MEAN
0616	3024		DCA X
0617	4247		JMS INIT / INITIALIZE
0620	1443	LOOP,	TAD I YLOC
0621	7010		RAR
0622	7012		RTR / CORRECT FOR HISTOGRAM RISE INCREMENT
0623	0053		AND MASKV / 0177
0624	1024		TAD X
0625	7510		SPA / COMPUTE (X-MEAN) ↑ 2
0626	7041		CIA
0627	3277		DCA SS
0630	1277		TAD SS
0631	4275		JMS SQ
0632	4263		JMS SUM / SQUARE & SUM
0633	2043		ISZ YLOC
0634	2043		ISZ YLOC
0635	2021		ISZ N01
0636	5220		JMP LOOP
0637	1072		TAD LOR
0640	7421		MQL / DISPLAY 24 BIT UNAVERAGED VARIANCE
0641	1071		TAD HOR / HIGH ORDER IN ACC
0642	7402		HLT / LOW ORDER IN MQ
0643	1077		TAD NPT2 / DISPLAY NUMBER OF
0644	7421		MQL / SAMPLES (24 BITS)
0645	1076		TAD NPT1 / HIGH ORDER IN ACC
0646	7402		HLT / LOW ORDER IN MQ
0647	0000	INIT,	0 / INITIALIZATION ROUTINE
0650	7300		CLA CLL
0651	3071		DCA HOR
0652	3072		DCA LOT
0653	3067		DCA HIGH
0654	3070		DCA LOW



0655	1311		TAD N00
0656	7041		CIA
0657	3021		DCA N01
0660	1041		TAD Y0
0661	3043		DCA YLOC
0662	5647		JMP I INIT
0663	0000	SUM,	0 / 24 BIT ADDER
0664	7100		CLL / ENTER WITH NO. IN HIGH & LOW
0665	1070		TAD LOW
0666	1072		TAD LOR
0667	3072		DCA LOR
0670	7204		GLK
0671	1071		TAD HOR
0672	1067		TAD HIGH
0673	3071		DCA HOR
0674	5663		JMP I SUM
0675	0000	SQ,	0 / 24 BIT SQUARING ROUTINE
0676	7425		MQL MUY
0677	0000	SS,	0
0700	3067		DCA HIGH
0701	7501		MQA
0702	3070		DCA LOW
0703	5675		JMP I SQ / EXIT WITH PRODUCT IN HIGH & LOW
0704	0000	AV,	0 / 24 BIT DIVIDER
0705	1072		TAD LOR
0706	7421		MQL
0707	1071		TAD HOR
0710	7407		DVI
0711	1777	N00,	1777 / 1024 QUANTIZATION LEVELS
0712	7701		CLA MQA
0713	5704		JMP IAV / EXIT WITH 12 BIT QUOTIENT IN ACC

